
Es mag ungewöhnlich erscheinen, ein Fachbuch zu schreiben, das nur einen Teilbereich von 
zwei so vielseitigen Programmen wie Excel und Power BI Desktop behandelt. Warum nicht 
stattdessen ein Buch, das entweder Excel oder Power BI in allen Facetten erklärt?
Wenn Sie dieses Buch gekauft haben, ahnen Sie zumindest, welches Potenzial in den Fea
tures schlummert, die unter dem Namen Power Query zusammengefasst werden. ExcelNut
zer verbringen unzählige Stunden damit, Daten aus den verschiedensten Quellen zusam
menzukopieren und dabei händisch zu modellieren. Etliche PowerBINutzer wissen zwar, 
wie Daten abgerufen werden, stoßen aber schnell an ihre Grenzen, wenn die Datenbasis 
nicht richtig aufbereitet ist. Wenn Sie nicht zu diesen Benutzern gehören wollen, haben Sie 
das richtige Buch gefunden.

Das Thema des vorliegenden Buchs
Was als AddIn für Excel begann, ist nun fester Teil von einer wachsenden Menge an 
MicrosoftProdukten, allen voran Excel und Power BI Desktop. Die Funktion der Abfragen 
und die Benutzeroberfläche des AbfrageEditors sind dabei bis auf wenige Ausnahmen pro
grammübergreifend gleich. Der Hauptunterschied ist die Hintergrundfarbe (hellgrau bei 
Excel und dunkelgrau bei Power BI). Deshalb ist es durchaus möglich, Power Query sowohl 
für Excel als auch Power BI Desktop zu erklären.
Aus genau diesem Grund findet aber auch eine starke Beschränkung der abgedeckten The
men statt: Das Buch behandelt nur das Abrufen von externen Daten und die Bearbeitung 
(Transformation) im AbfrageEditor, basierend auf der Abfragesprache M. Sie werden ler
nen, wie Sie Daten in Ihr Programm laden und dabei genau in die Form bringen, in der Sie 
sie benötigen. Die so aufbereiteten Tabellen können dann in Excel und Power BI nach Belie
ben interpretiert und visualisiert werden – doch das ist nicht mehr Thema des Buchs.

Unterschiede zwischen den Versionen
Die Grundlagen der AbfrageLogik und der Formelsprache M haben sich seit den ersten 
Versionen von Power Query kaum verändert. Die Benutzeroberfläche und die Entwicklungs
umgebung des AbfrageEditors werden jedoch laufend optimiert.
Die regelmäßige Verbesserung eines Programms ist natürlich zu begrüßen, doch erschwert 
sie das Lernen und Lehren der Grundlagen. Es ist nicht auszuschließen, dass das eine oder 
andere Beispiel auf Ihrem Computer nicht genauso funktioniert wie hier beschrieben – sei 

es, dass Sie eine ältere Version benutzen oder sei es, dass nach der Buchveröffentlichung 
weitere Neuerungen eingeführt wurden.
Seien Sie aber bitte nicht frustriert, wenn eine Beschreibung oder eine Abbildung nicht 
exakt zutrifft. In den meisten Fällen ist das entsprechende Werkzeug nur anders benannt 
oder befindet sich an einer anderen Stelle. Spätestens wenn es um das Schreiben von 
MCode geht, sind Versionsunterschiede ohnehin kaum mehr von Belang.

Kontakt zum Autor
Auch bei einem so starren Medium wie einem Buch darf der Kontakt zwischen Leser und 
Autor nicht zu kurz kommen. Scheuen Sie sich nicht, mir zu schreiben, wenn Sie Fragen zu 
einem der behandelten Themen haben. Ein SoftwareSpezialist lebt von solchen Herausfor
derungen. Feedback jeder Art, egal ob Lob oder Kritik, ist natürlich auch immer willkom
men. Schicken Sie einfach eine Mail an ignaz.alexander@schels.de.



Mit dem Verständnis der MGrundlagen eröffnen sich neue Möglichkeiten, um Abfragen 
flexibler und sicherer zu gestalten. Die meisten Abfragen beruhen auf festen Annahmen: 
Die Datei bzw. Datenbank muss an einem vordefinierten Ort liegen. Ihr Inhalt muss exakt 
dasselbe Layout haben wie zu dem Zeitpunkt, als die Abfrage erstellt wurde. Die Abfrage 
arbeitet Schritt für Schritt, wobei sich jeder Schritt auf den vorherigen bezieht.
Obwohl diese Aussagen auf mindestens 90 % aller Abfragen zutreffen, sind sie keineswegs 
zwingend notwendig. Mit geschickten Anpassungen des MCodes können Sie starre Abfra
gen auf die jeweiligen Bedürfnisse anpassen.

 ■ 13.1  Nicht-lineare Abfragen

Die meisten Programmiersprachen beinhalten das Konzept von Variablen: Ein Zeichen oder 
Wort kann als Platzhalter für einen Wert dienen und bei Bedarf als Referenz auf diesen Wert 
benutzt werden. Das, was in diesem Buch als Schrittnamen bezeichnet wird, sind im Prin
zip Variablen: Ihnen wird ein Wert zugewiesen, damit sie später an Stelle dieses Werts 
verwendet werden können.
Werfen Sie ein Blick auf den Code in Listing 13.1:

Listing 13.1 Eine einfache Abfrage auf eine Excel-Datei.

let
    Quelle = Excel.Workbook(File.Contents(
             “C:\Beispiele\13-01-Pegel-Isar.xlsx”), null, true),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet, 
             [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes(#“Höher gestufte Header”,
             {{“Datum”, type date}, {“Uhrzeit”, type time},
             {“Wasserstand cm über NN”, Int64.Type}})
in
    #“Geänderter Typ”

Die Abfrage besteht aus vier Schritten: Quelle, Tabelle1_Sheet, Höher gestufte Header 
und Geänderter Typ. Mit Ausnahme des ersten Schritts finden sich in allen Wertzuweisun

gen Verweise auf den vorhergehenden Schritt. Diese Reihenfolge erscheint zwar logisch, ist 
aber für das Ergebnis völlig irrelevant. Die Abfrage würde genauso funktionieren, wenn die 
Schritte in einer anderen Reihenfolge geschrieben wären.

Listing 13.2 Die Transformationsschritte werden vertauscht.

let
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet, 
             [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes(#“Höher gestufte Header”,
             {{“Datum”, type date}, {“Uhrzeit”, type time},
             {“Wasserstand cm über NN”, Int64.Type}}),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    Quelle = Excel.Workbook(File.Contents(
             “C:\Beispiele\13-01-Pegel-Isar.xlsx”), null, true)
in
    #“Geänderter Typ” 
 

Der Befehl Quelle, der im ursprünglichen Code am Anfang stand, steht nun am Ende der 
Abfrage. Der erste Transformationsschritt Höher gestufte Header kann natürlich nicht 
aus geführt werden, bevor die Verbindung zur Quelldatei hergestellt wurde. Dennoch liefert 
die Abfrage in Listing 13.2 dasselbe Ergebnis wie die OriginalAbfrage aus Listing 13.1. Power 
Query ist also im Stande, die sinnvollste Reihenfolge der Abarbeitung selbst zu ermitteln.
Diese Erkenntnis kann von großem Nutzen sein, denn sie ermöglicht es, Abfragen zu 
schreiben, die nicht linear ablaufen, sondern Abzweigungen oder sogar Schleifen enthalten.
Betrachten wir folgendes Beispiel, das einen einfachen, aber häufigen Fall zeigt. 

Bild 13.1  Die Struktur der Beispieldatei.

Die Beispieldatei 13-02-Export_Feb.txt enthält eine Tabelle, deren Spalten durch TabZeichen 
getrennt sind. Allerdings gibt es vor und nach der Tabelle Zeilen, die nicht zur Tabelle ge 
hören. Das Problem ist, dass sich die Anzahl der Zeilen vor der Tabelle sowie die Anzahl der 
Zeilen in der Tabelle ändern können. Um die Tabelle vom Rest isolieren zu können, kann 
also keine fixe Zeilenzahl angenommen werden. Hinzu kommt, dass die Abteilungsnummer 
in den Kopfzeilen in die jeweiligen Tabellenzeilen übernommen werden sollte. Sie muss 
also irgendwie gespeichert werden, bevor die Kopfzeilen entfernt werden.

 � Erstellen Sie eine Abfrage auf die Textdatei 13-02-Export_Feb.txt und bearbeiten Sie sie im 
AbfrageEditor. Stellen Sie dabei sicher, dass als Trennzeichen Tab eingestellt ist.

 � Löschen Sie den automatisch eingefügten Transformationsschritt Geänderter Typ, er ist 
hier überflüssig.

 � Zunächst soll die Abteilungsnummer in der vierten Zeile extrahiert werden. Filtern Sie 
hierfür die erste Spalte nach Werten beginnend mit “Abteilung:”.

Ein normaler Textfilter wäre hier problematisch. Wenn Sie die Filterfunktion 
 benutzen und in das Textfeld das Wort “Abteilung” eingeben, sieht es zwar so 
aus, als würde nur nach diesem Begriff gefiltert.
Ein Blick auf die Bearbeitungsleiste zeigt jedoch, dass der Befehl den kompletten 
Wert „Abteilung: P2“ enthält, auch wenn Sie nur nach „Abteilung“ gesucht haben.

Bild 13.2  Der Befehl in der Bearbeitungsleiste.

Damit auch bei anderen Abteilungen der richtige Wert extrahiert wird, müssen 
Sie daher den Textfilter Beginnt mit . . . verwenden.



 � Da wir nur die Abteilungsnummer, hier „P2“ benötigen, muss der Rest des Werts entfernt 
werden. Klicken Sie dafür mit der rechten Maustaste auf die erste Spalte, wählen Sie 
Werte ersetzen und ersetzen Sie die Zeichenfolge „Abteilung: “ (mit Doppelpunkt und 
Leerzeichen) durch nichts.

 � Der momentan angezeigte Wert ist eine Tabelle. Es fehlt aber nur noch ein Schritt, um den 
gesuchten Textwert zu isolieren: Klicken Sie mit der rechten Maustaste auf den Zellwert 
P2 und wählen Sie im Kontextmenü den Eintrag Drilldown ausführen.

Bild 13.3  Ein Drilldown wählt einen einzelnen Wert aus einer Tabelle.

In der Bearbeitungsleiste sehen Sie, wie der zugehörige MBefehl aussieht:

= #“Ersetzter Wert”{0}[Column1]

Aus der Tabelle, die der vorherige Schritt zurückgegeben hat, wurde der Wert in der ersten 
Zeile (d. h. der Eintrag 0 in der Liste) und der ersten Spalte (mit dem Namen Column1) aus
gewählt. Der aktuelle Schritt gibt also den Textwert P2 zurück.

 � Benennen Sie den aktuellen Transformationsschritt um in Abteilung.
 � Wie kommen Sie nun wieder an die Ausgangstabelle? Ganz einfach: Erstellen Sie einen 
neuen Transformationsschritt über das fxSymbol in der Bearbeitungsleiste und geben 
Sie als Code für diesen Schritt Folgendes ein:

= Quelle

 � Bestätigen Sie mit Enter. Da Quelle der Name des ersten Schritts war, erscheint nun der 
Rückgabewert dieses Schritts wieder.

Auf ähnliche Weise sollen nun die Zeilennummer für den Beginn und das Ende der Tabelle 
extrahiert werden. Der besseren Übersicht wegen, reduzieren wir die Tabelle auf eine 
Spalte. Sie wissen ja nun, wie Sie die OriginalTabelle bei Bedarf wieder zurückholen.

 � Entfernen Sie alle Spalten außer der ersten über den Befehl Andere Spalten entfernen.
 � Klicken Sie im Register Spalte hinzufügen auf Indexspalte.

Bild 13.4  Die Indexspalte liefert Zeilennummern, beginnend mit 0.

 � Um an den Beginn der gesuchten Tabelle zu kommen, filtern Sie die erste Spalte nach 
Werten, die mit dem Text „BestellNr“ beginnen.

 � Führen Sie einen Drilldown auf die Zeilennummer in der Indexspalte aus.
 � Benennen Sie den letzten Abfrageschritt um in Anfang.
 � Um die Abfrage etwas zu verkürzen, können Sie die letzten beiden Schritte zu einem 
zusammenfassen. Öffnen Sie dazu den Erweiterten Editor.

 � Der vorletzte Abfrageschritt sollte Gefilterte Zeilen1 heißen. Markieren Sie den zugehö
rigen Ausdruck rechts vom Gleichheitszeichen bis vor das Komma am Zeilenende. Kopie
ren Sie den markierten Text in den Transformationsschritt Anfang und ersetzen Sie dort 
den Namen des alten Schritts #“Gefilterte Zeilen1”. Der Befehl sollte nun so aussehen:

Anfang = Table.SelectRows(#“Hinzugefügter Index”, each 
    Text.StartsWith([Column1], “Bestell-Nr”)){0}[Index]

 � Anschließend können Sie den Schritt Gefilterte Zeilen1 löschen.
Für das Ende der Tabelle soll die gleiche Logik angewandt werden. Nun, da Sie alles, was 
dafür nötig ist, in einem Schritt haben, können Sie den Befehl direkt im Editor schreiben.

 � Fügen Sie unter dem Schritt Anfang einen neuen Schritt Ende ein. Der Code dafür ist fast 
identisch, nur statt “Bestell-Nr” lautet der FilterText Bestellungen. Der Schritt Ende 
hat also diesen Code:

Ende = Table.SelectRows(#“Hinzugefügter Index”, each 
    Text.StartsWith([Column1], “Bestellungen”)){0}[Index]

 � Geben Sie ein Komma am Ende des Schritts Anfang ein.
 � Ersetzen Sie den Schrittnamen in der letzten Zeile nach dem in durch Ende.
 � Schließen Sie den Erweiterten Editor.
 � Fügen Sie einen neuen Schritt ein und geben Sie als Code dafür = Quelle ein. Somit 
erscheint wieder die Ursprungstabelle.

Die Werte für Abteilung, Anfang und Ende können nun in weiteren Transformationen ver
wendet werden.

 � Klicken Sie im Register Start auf Zeilen beibehalten – Erste Zeilen beibehalten.
 � Geben Sie als Anzahl die Zahl 10 ein – sie ist nur ein Platzhalter und wird gleich ersetzt.
 � In der Bearbeitungsleiste sehen Sie den Code, den Sie damit generiert haben:

= Table.FirstN(Benutzerdefiniert2, 10)

 � Anstelle der Zahl 10 soll die vorhin ermittelte Zeilenanzahl stehen. Ersetzen Sie die 10 
durch Ende.

 � Der Schritt Benutzerdefiniert2 verweist eigentlich nur auf den ersten Schritt Quelle. Er ist 
daher überflüssig. Ersetzen Sie Benutzerdefiniert2 im letzten Befehl durch Quelle. Der 
Befehl sollte nun also so aussehen:

= Table.FirstN(Quelle, Ende)

 � Bestätigen Sie mit Enter und löschen Sie den Schritt Benutzerdefiniert2.
 � Klicken Sie im Register Start auf Zeilen entfernen – Erste Zeilen entfernen. Geben Sie 
eine beliebige Zahl ein, zum Beispiel 3.

 � Die eingegebene Zahl wird nun in der Bearbeitungsleiste ersetzt durch Anfang.
 � Klicken Sie im StartRegister auf Erste Zeile als Überschriften verwenden. Die Tabelle 
ist nun isoliert.

 � Zum Abschluss wird noch die Abteilungsnummer angefügt. Klicken Sie im Register 
Spalte hinzufügen auf Benutzerdefinierte Spalte.

 � Als Formel für die neue Spalte geben Sie einfach den Namen des Transformationsschritts 
an, in dem der Wert gespeichert wurde. Geben Sie also sowohl als Formel als auch als 
Spaltentitel Abteilung ein.

Bild 13.5  Die Tabelle ist nun abgegrenzt.

Damit ist die Transformation abgeschlossen. Sie haben die Tabelle aus dem Dateiinhalt 
isoliert, ohne wichtige Informationen zu verlieren und ohne feste Zeilenangaben zu verwen
den. Die Abfrage ist daher flexibel, was die Position und Höhe der Tabelle angeht. Sie kön
nen es ausprobieren, indem Sie im Transformationsschritt Quelle den Pfad für die Beispiel
datei 13-03-Export_Mrz.txt eingeben. Auch hier werden die richtigen Zeilen eingelesen.

 ■ 13.2  Unterschiedliche Tabellenzeilen 
 vergleichen

Es wurde bereits gezeigt, wie neue Spalten an Tabellen angefügt werden, in denen Berech
nungen mit Werten aus anderen Spalten angestellt werden. Doch was, wenn sich die Berech
nung auf unterschiedliche Zeilen innerhalb der Tabelle beziehen soll?

Bild 13.6  Ausschnitt aus einer Zeiterfassungsliste.

Bild 13.6 zeigt einen Ausschnitt aus einer Beispieldatei, die Schichtzeiten mehrerer Mit
arbeiter enthält. Das Ziel ist, eine zusammenhängende Tabelle mit den Arbeitsstunden der 
Mitarbeiter anhand ihrer jeweiligen Personalnummer zu erzeugen. Dabei stellen sich zwei 
Herausforderungen:
1. Die Personalnummer der jeweiligen Mitarbeiter steht über den einzelnen Tabellen. Das 

Wort „PersonalNr“ steht aber in einer anderen Zeile, als die Nummer selbst. Die Num
mer kann also nur anhand der vorhergehenden Zeile gefunden werden.

2. Die Zeiten für Schichtbeginn und Schichtende stehen in unterschiedlichen Zeilen. Um 
die Stundenzahl zu berechnen, muss also die Differenz aus untereinanderliegenden Wer
ten ermittelt werden.

Wie schon im letzten Beispiel besteht der „Trick“ auch hier in der Verwendung einer Index
spalte. Sie ermöglicht es, eine bestimmte Zeilennummer aus der Tabelle auszuwählen.

 � Erstellen Sie eine neue Abfrage auf die ExcelDatei 13-04-Zeiterfassung.xlsx.
 � Fügen Sie eine Indexspalte hinzu, indem Sie im Register Spalte hinzufügen auf 
 Indexspalte klicken. Achten Sie darauf, dass der Index mit 0 beginnt.

 � Ändern Sie den Namen des gerade erstellten Transformationsschritts in neuIndex. Wenn 
der Schrittname kein Leerzeichen enthält, lässt er sich leichter in Formeln verwenden 
(man spart sich das #“...”).

 � Fügen Sie als Nächstes eine Benutzerdefinierte Spalte hinzu. Geben Sie als Spaltentitel 
Personal-Nr. ein und als Spaltenformel folgenden Code:

 if [Column1] = “Personal-Nr.:” then
    neuIndex[Column1]{[Index]+1}
 else
    null

Bild 13.7  Die Personalnummer wird über den Index eingefügt.

Der Transformationsschritt neuIndex wird hier verwendet, um einen Wert aus der entspre
chenden Tabelle zu extrahieren. In den eckigen Klammern wird die gesuchte Spalte angege
ben, in den geschweiften Klammern die Zeilennummer. Anstelle der Zeilennummer wird 
hier [Index]+1 angegeben. Da [Index] keinen Schrittnamen vorangestellt hat, bezieht es 
sich auf die Spalte im aktuellen Schritt und da wir uns innerhalb der Formel für eine benut
zerdefinierte Spalte befinden, bezieht es sich auf den Wert der jeweiligen Zeile. Die Zeilen
nummer wird um eins erhöht, um die darunterliegende Zeile auszulesen.

 � Wenden Sie die Funktion Ausfüllen – Nach unten auf die neu hinzugefügte Spalte an.
 � Die Kopf und Leerzeilen können nun entfernt werden. Filtern Sie die zweite Spalte so, 
dass alle NullWerte entfernt werden.

 � Klicken Sie im StartRegister auf Erste Zeile als Überschriften verwenden.
 � Durch den letzten Schritt gingen die Titel der letzten beiden Spalten verloren. Ändern Sie 
die Überschriften wieder auf Index und Personal-Nr..

 � Es müssen noch einige Zwischenüberschriften entfernt werden. Filtern Sie die Spalte 
Pausen, indem Sie dort das Wort Pausen entfernen.

 � Entfernen Sie den Text „ Min.“ (mit Leerzeichen und Punkt) aus der Spalte Pausen mit
hilfe von Werte ersetzen.

 � Ändern Sie die Datentypen der Spalten: Datum zu Datum, Uhrzeit zu Zeit, Pausen zu ganze 
Zahl und Personal-Nr. zu Text.

Bild 13.8  Die Tabelle ist nun abgegrenzt.

Nun, da die Tabelle ordentliche Überschriften und keine Leerräume mehr hat, kann die 
letzte Herausforderung in Angriff genommen werden: Anstelle der Uhrzeiten für Kommen 
und Gehen, soll die Anzahl der Stunden für jede Arbeitseinheit errechnet werden. Es muss 
also die Differenz aus den Zeiten Kommt und Geht abzüglich der PausenMinuten errechnet 
werden.

 � Nachdem durch die Filteraktionen mehrere Zeilen entfernt wurden, ist die Indexspalte 
nicht mehr fortlaufend und dadurch unbrauchbar. Entfernen Sie sie und erstellen Sie eine 
neue Indexspalte.

 � Ändern Sie den Namen des letzten Transformationsschritts zu neuIndex2.
 � Fügen Sie eine neue Benutzerdefinierte Spalte hinzu. Geben Sie als Namen der neuen 
Spalte Dauer mit Pause ein. Die Spaltenformel lautet folgendermaßen:

 if [Aktion] = “geht” then
    [Uhrzeit] - neuIndex2[Uhrzeit]{[Index]-1}
 else
    null

 � Ändern Sie den Spaltentyp in Dauer.
 � Filtern Sie alle NullWerte heraus.
 � Entfernen Sie die Spalten Uhrzeit, Aktion und Index.
 � Fügen Sie eine neue Benutzerdefinierte Spalte hinzu. Geben Sie den Namen Dauer ohne 
Pause und folgende Spaltenformel ein:

 [Dauer mit Pause] - #duration(0, 0, [Pausen], 0)

Das Ergebnis kann sich sehen lassen: Anstelle der Uhrzeiten wird nun die Dauer jeder 
Schicht im Format Duration angezeigt.

Bild 13.9  Die finale Tabelle.

Wenn Sie anstelle des Formats Duration lieber die Stunden als Zahl anzeigen 
möchten, multiplizieren Sie die Dauer mit 24.



 ■ 13.3  Abfragen mit Parametern

Abfragen mit Power Query sind, nachdem sie einmal erstellt wurden, kinderleicht zu bedie
nen: Im Prinzip muss man nur wissen, wie sie aktualisiert werden. Diese Einfachheit kann 
aber auch ein Nachteil sein, wenn man den Ablauf der Abfrage „von außen“ beeinflussen 
möchte, d. h. ohne den AbfrageCode zu ändern.
Stellen Sie sich vor, Sie haben eine grandiose Abfrage auf eine Datei erstellt und möchten 
sie mit all Ihren Kollegen teilen. Nun möchten diese Kollegen aber unterschiedliche Quell
dateien verwenden. Es gibt nun drei Möglichkeiten, um ihnen zu helfen: Sie zeigen jedem 
Kollegen, wie der entsprechende Transformationsschritt im AbfrageEditor angepasst wird. 
Oder Sie bitten sie, die Quelldateien genauso zu benennen und an genau den Ordnerpfad zu 
kopieren, wie es in der Abfrage angegeben ist. Oder – und das ist vermutlich der Weg, mit 
dem Sie am wenigsten Verwirrung stiften – Sie fügen einen Platzhalter in den Abfragecode 
ein, dessen Wert relativ einfach angepasst werden kann.

13.3.1  Power-Query-Parameter

 � Erstellen Sie als Erstes eine normale Abfrage auf die Datei 13-05-Pegel-Donau.xlsx.
 � Falls die Datentypen nicht richtig erkannt werden, stellen Sie sie entsprechend ein.

Der Abfragecode sollte nun in etwa so aussehen:

Listing 13.3 Der Abfragecode ohne Parameter.

let
    Quelle = Excel.Workbook(
         File.Contents(“C:\Beispiele\13-05-Pegel-Donau.xlsx”), null, true),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet, 
         [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes(
         #“Höher gestufte Header”,{{“Datum”, type date}, 
         {“Uhrzeit”, type time}, {“Wasserstand cm über NN”, Int64.Type}})
in
    #“Geänderter Typ”

Im ersten Transformationsschritt Quelle ist der Pfad der Datei enthalten, auf die Sie die 
Abfrage gemacht haben. Das ist der Teil, der angepasst werden muss, damit verschiedene 
Dateien an beliebigen Orten eingelesen werden können.

 � Im StartRegister gibt es das Menü Parameter verwalten. Wählen Sie den Menüeintrag 
Neuer Parameter.

 � Es erscheint ein Dialogfenster, in dem Sie die Eigenschaften Ihres Parameters eingeben 
können. Stellen Sie Folgendes ein:

 � Name: Dateipfad
 � Beschreibung: Pfad und Dateiname der Quelldatei
 � Erforderlich: Aktivieren (bewirkt, dass Nutzer eine Warnung erhalten, wenn das ent
sprechende Feld leer ist)

 � Typ: Text
 � Vorgeschlagene Werte: beliebiger Wert
 � Aktueller Wert: Geben Sie hier Ihren Dateipfad und Dateinamen ein.

Bild 13.10  Das Dialogfenster zur Erstellung des Parameters.

 � Bestätigen Sie mit OK.
Daraufhin wird der Parameter in der Liste der Abfragen angezeigt, wo er auch angepasst 
werden kann.

 � Wählen Sie in der Abfragenliste wieder die ursprüngliche Abfrage und öffnen Sie den 
Erweiterten Editor.

 � Ersetzen Sie im ersten Abfrageschritt den Dateipfad durch den Parameternamen. Der 
neue Abfragecode ist in Listing 13.4 zu sehen:

Listing 13.4 Der Abfragecode mit Parameter.

let
    Quelle = Excel.Workbook(
         File.Contents(Dateipfad), null, true),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet, 
         [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes(
         #“Höher gestufte Header”,{{“Datum”, type date}, 
         {“Uhrzeit”, type time}, {“Wasserstand cm über NN”, Int64.Type}})
in
    #“Geänderter Typ”

 � Geben Sie der Abfrage den Namen AbfrageWasserstand und schließen Sie sie ab.
Der Pfad der Quelldatei ist nun nicht fest im Abfragecode verankert, sondern nur als Para
meterwert hinterlegt. Er kann ganz einfach geändert werden, ohne dass man sich mit Power 
Query auskennen muss.

EXCEL:
 � Öffnen Sie das Menü für Abfragen und Verbindungen durch die entsprechende Schalt
fläche im Register Daten.

 � Durch Doppelklick auf den Eintrag für den Parameter Dateipfad gelangen Sie in den 
AbfrageEditor. Hier können Sie den Parameterwert direkt ändern.

Bild 13.11  Ändern des Parameterwerts in Excel.

 � Geben Sie nun den Dateipfad für die Beispieldatei 13-06-Pegel-Main.xlsx ein.
 � Bestätigen Sie mit Schliessen & Laden.
 � Aktualisieren Sie die Abfrage. Es wird nun eine neue Datei eingelesen.

POWER BI: 
In Power BI ist die Anpassung noch einfacher als in Excel. Sie müssen nicht einmal den 
AbfrageEditor öffnen, um den Wert eines Parameters zu ändern.

 � Klicken Sie im Hauptfenster (nicht im AbfrageEditor) im Register Start auf die Schalt
fläche Abfragen bearbeiten und wählen Sie den Menüeintrag Parameter bearbeiten.

Bild 13.12  In Power BI Desktop haben Sie im Hauptfenster Zugriff auf Parameter.

 � Geben Sie den Dateipfad für die Beispieldatei 13-06-Pegel-Main.xlsx ein.
 � Aktualisieren Sie die Abfrage. Es wird nun eine neue Datei eingelesen.

13.3.2  Parameter aus anderen Quellen

Alternativ zum Erstellen eines Parameters können Sie auch Werte aus anderen Quellen 
einlesen, um damit Ihre Abfrage zu steuern. Eine einfache ExcelTabelle kann beispiels
weise dafür verwendet werden, den Dateipfad einer Quelldatei oder die Filteroptionen einer 
Tabelle einzustellen. Diese Technik ist für ExcelNutzer besonders vorteilhaft, denn Sie kön
nen Ihre selbstdefinierten Parameter in der gleichen Datei speichern wie die dazugehörige 
Abfrage. Für Nutzer von Power BI ist dies zwar nicht möglich, die zugrunde liegende Tech
nik funktioniert aber trotzdem.
Die Beispieldatei 13-07-Parameterabfrage.xlsx enthält eine einfache Tabelle, in der Parame
ter und die zugehörigen Werte eingetragen sind (vgl. Bild 13.13). Diese Tabelle kann benutzt 
werden, um eine Abfrage auf eine Datei zu verändern.

Bild 13.13  Die Parameter liegen in einer Excel-Tabelle.

 � Öffnen Sie die Datei 13-07-Parameterabfrage.xlsx
 � Tragen Sie in der rechten Spalte den Pfad ein, an dem Ihre Beispieldateien liegen.

EXCEL:
 � Markieren Sie eine Zelle in dieser Tabelle und erstellen Sie eine neue Abfrage auf die 
Tabelle. Klicken Sie hierfür im Register Daten auf Daten abrufen  – Aus anderen 
 Quellen – Aus Tabelle / Bereich.

 � Löschen Sie den automatisch eingefügten Transformationsschritt Geänderter Typ.
 � Klicken Sie mit der rechten Maustaste auf den ersten Wert in der rechten Spalte und wäh
len Sie den Befehl Drilldown ausführen.

 � Öffnen Sie den Erweiterten Editor.
 � Fassen Sie die beiden Transformationsschritte zu einem Schritt zusammen. Listing 13.5 
zeigt die so entstehende Anweisung.

Listing 13.5 Der Befehl zum Auslesen eines Werts aus der Tabelle.

    Quelle = Excel.CurrentWorkbook(){[Name=“tbl_Parameter”]}[Content]{0}[Wert]

 � Nun wissen Sie, welcher Befehl nötig ist, um einen Wert aus der Parametertabelle zu 
extrahieren. Sie werden diesen Befehl später brauchen. Die Abfrage selbst ist nicht mehr 
notwendig. Beenden Sie den AbfrageEditor, ohne zu speichern.

 � Erstellen Sie nun eine Abfrage auf die Datei 13-12-Einwohnerzahlen_2010.xlsx und be 
arbeiten Sie sie im AbfrageEditor.

 � Filtern Sie die Spalte Stadt nach einem einzelnen Wert. Welche Stadt Sie wählen, ist egal, 
weil sie später durch den Parameterwert ersetzt wird.

 � Öffnen Sie den Erweiterten Editor. Hier werden nun die Teile des Codes, die flexibel sein 
sollen (Dateipfad und Stadt) an den entsprechenden Stellen eingesetzt.
Der erste Transformationsschritt sieht zunächst so aus (wobei natürlich der Pfad auf 
Ihrem System unterschiedlich sein wird):

Quelle = Excel.Workbook(File.Contents(
    “C:\Beispiele\13-12-Einwohnerzahlen_2010.xlsx”), null, true),

 � Anstelle des eingetragenen Dateipfads in Anführungszeichen fügen Sie den Befehl ein, 
der den Dateipfad aus der Parametertabelle liest (siehe oben). Der neue Befehl sieht somit 
folgendermaßen aus:

Quelle = Excel.Workbook(File.Contents(
 Excel.CurrentWorkbook(){[Name=“tbl_Parameter”]}[Content]{0}[Wert]
 ), null, true),

 � Setzen Sie ebenso für die herausgefilterte Stadt den entsprechenden CodeBaustein ein. 
Dabei müssen Sie nur beachten, dass nun die zweite Zeile aus der Tabelle ausgelesen 
wird. Die Zahl in Klammern muss somit nicht 0, sondern 1 sein.

Der Abfragecode sollte am Ende so aussehen wie in Listing 13.6.

Listing 13.6 Der Abfragecode mit den eingesetzten Parameterabfragen.

let
    Quelle = Excel.Workbook(File.Contents(
        Excel.CurrentWorkbook(){[Name=“tbl_Parameter”]}[Content]{0}[Wert]
        ), null, true),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet,
        [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes
        (#“Höher gestufte Header”,{{“Stadt”, type text},
        {“Einwohnerzahlen”, Int64.Type}}),
    #“Gefilterte Zeilen” = Table.SelectRows(#“Geänderter Typ”, each ([Stadt] = 
        Excel.CurrentWorkbook(){[Name=“tbl_Parameter”]}[Content]{1}[Wert]
        ))
in
    #“Gefilterte Zeilen”

 � Schließen Sie die Abfrage nun ab. Wählen Sie dabei am besten Schliessen & Laden in . . . 
und legen Sie fest, dass die Abfragetabelle direkt unter der Parametertabelle angezeigt 
wird. So haben Sie alles im Blick.

Bild 13.14  Flexible Datei- und Filterwahl in einem Excel-Blatt.

 � Testen Sie Ihre Abfrageparameter, indem Sie verschiedene Städte als Parameter eingeben 
und die Abfrage aktualisieren. Sofern die gesuchte Stadt in der Liste ist, sollte die ent
sprechende Einwohnerzahl erscheinen.

 � Geben Sie unterschiedliche Dateipfade an, um die Einwohnerzahlen aus verschiedenen Jah
ren zu erhalten. Wählen Sie eine dieser Dateinamen (zusammen mit dem jeweiligen Pfad):

 � 1308Einwohnerzahlen_1970.xlsx
 � 1309Einwohnerzahlen_1980.xlsx
 � 1310Einwohnerzahlen_1990.xlsx
 � 1311Einwohnerzahlen_2000.xlsx
 � 1312Einwohnerzahlen_2010.xlsx

Mit ein wenig Excel-Bastelei können Sie Ihre Parametertabelle noch benutzer-
freundlicher machen. Sie können zum Beispiel mithilfe der Datenüberprüfung 
auf dem Register Daten ein Dropdown einfügen, mit dem man die Stadt aus-
wählen kann. Werfen Sie hierfür einen Blick auf die Datei 13-13-Parameter-
abfrage_Lösung.xlsx.



POWER BI: 
In Power BI können Sie ebenfalls Abfragen erstellen, deren Parameter in einer ExcelTabelle 
liegen. Im Unterschied zum obigen Beispiel wäre die Parametereingabe dann jedoch in 
einer anderen Datei als die Abfrage, was eher selten zweckmäßig ist.
Nichtsdestotrotz ist es natürlich möglich. Das Vorgehen ist das gleiche wie oben beschrie
ben, mit dem Unterschied, dass die Funktion Excel.CurrentWorkbook in Power BI kein Ergeb
nis liefert. Stattdessen müssen Sie die ExcelDatei über die Funktion Excel.Workbook ein
lesen. Der CodeBaustein, der den ersten Parameter aus der Tabelle zurückgibt, lautet 
folgendermaßen:

Excel.Workbook(File.Contents
    (“C:\Beispiele\13-07-Parameterabfrage.xlsx”), null, true)
    {[Item=“tbl_Parameter”,Kind=“Table”]}[Data]{0}[Wert]

Dabei müssen Sie natürlich die Pfadangabe in der zweiten Zeile durch den richtigen Pfad 
auf Ihrem System ersetzen.
Der komplette Code für Power BI wird in Listing 13.7 gezeigt.

Listing 13.7 Der Abfragecode mit Parametern aus einer externen Excel-Datei.

let
    Quelle = Excel.Workbook(File.Contents(
        Excel.Workbook(File.Contents
        (“C:\Beispiele\13-07-Parameterabfrage.xlsx”), null, true)
        {[Item=“tbl_Parameter”,Kind=“Table”]}[Data]{0}[Wert]        
        ), null, true),
    Tabelle1_Sheet = Quelle{[Item=“Tabelle1”,Kind=“Sheet”]}[Data],
    #“Höher gestufte Header” = Table.PromoteHeaders(Tabelle1_Sheet,
        [PromoteAllScalars=true]),
    #“Geänderter Typ” = Table.TransformColumnTypes
        (#“Höher gestufte Header”,{{“Stadt”, type text},
        {“Einwohnerzahlen”, Int64.Type}}),
    #“Gefilterte Zeilen” = Table.SelectRows(#“Geänderter Typ”, each ([Stadt] = 
        Excel.Workbook(File.Contents
        (“C:\Beispiele\13-07-Parameterabfrage.xlsx”), null, true)
        {[Item=“tbl_Parameter”,Kind=“Table”]}[Data]{1}[Wert]  
        ))
in
    #“Gefilterte Zeilen”

